TLP 18 (3–4): 520–534, 2018.

C Cambridge University Press 2018


520

doi:10.1017/S1471068418000182

Routing Driverless Transport Vehicles in Car
Assembly with Answer Set Programming
M A R T I N G E B S E R, P H I L I P P O B E R M E I E R and
TORSTEN SCHAUB
University of Potsdam, Germany
(e-mail: gebser@cs.uni-potsdam.de)

M I C H E L R A T S C H - H E I T M A N N and M A R I O R U N G E
Mercedes-Benz Ludwigsfelde GmbH, Germany
submitted 30 April 2018; accepted 11 May 2018

Abstract
Automated storage and retrieval systems are principal components of modern production and
warehouse facilities. In particular, automated guided vehicles nowadays substitute humanoperated pallet trucks in transporting production materials between storage locations and assembly stations. While low-level control systems take care of navigating such driverless vehicles along
programmed routes and avoid collisions even under unforeseen circumstances, in the common
case of multiple vehicles sharing the same operation area, the problem remains how to set up
routes such that a collection of transport tasks is accomplished most eﬀectively. We address
this prevalent problem in the context of car assembly at Mercedes-Benz Ludwigsfelde GmbH,
a large-scale producer of commercial vehicles, where routes for automated guided vehicles used
in the production process have traditionally been hand-coded by human engineers. Such adhoc methods may suﬃce as long as a running production process remains in place, while any
change in the factory layout or production targets necessitates tedious manual reconﬁguration,
not to mention the missing portability between diﬀerent production plants. Unlike this, we
propose a declarative approach based on Answer Set Programming to optimize the routes
taken by automated guided vehicles for accomplishing transport tasks. The advantages include
a transparent and executable problem formalization, provable optimality of routes relative
to objective criteria, as well as elaboration tolerance towards particular factory layouts and
production targets. Moreover, we demonstrate that our approach is eﬃcient enough to deal with
the transport tasks evolving in realistic production processes at the car factory of Mercedes-Benz
Ludwigsfelde GmbH.
KEYWORDS: automated guided vehicle routing, car assembly operations, answer set programming

1 Introduction
Automated guided vehicles play a key role in modern industries, let it be in warehouses,
mines, or as in our case production facilities. Most of the time, however, these vehicles
are programmed by human engineers to execute speciﬁc tasks. This makes it impossible
to quickly reassign tasks in case of breakdowns or to easily react to changing production requirements, not to mention the missing portability between diﬀerent production
plants and factory layouts. The lack of elaboration tolerance does not only lead to high

Fig. 1. Real-world factory layout with transport corridors and directions indicated by arrows

expenditures, but the resulting rigid control also rules out any ﬂexible ﬂeet management.
In particular, no conclusions can be drawn about the eﬀectiveness or even optimality of
pre-programmed vehicle routes.
In view of these common circumstances, we address the challenge of devising a new
control system for automated guided vehicles supplying the assembly lines at the car
factory of Mercedes-Benz Ludwigsfelde GmbH, which is expected to be ﬂexible enough
to adapt to malfunctions and emerging requirements, and whose quality of operation can
be measured and optimized relative to given objectives. To make our speciﬁc task more
precise, consider the retouched layout of an assembly hall at the production plant of
Mercedes-Benz Ludwigsfelde GmbH in Figure 1. The overall goal is to guarantee that all
car components are at their designated place next to the assembly line when they are due
for installation. The corresponding transport tasks are accomplished by a heterogeneous
ﬂeet of automated guided vehicles that fetch the necessary components from several
storage areas. For instance, a vehicle may ﬁrst halt at some storage location to load
production material, then move on to an assembly station in need of the material, and
from there cart oﬀ leftover material to a recycling facility. The sketched task thus involves
stopovers at three distinct locations, between which the vehicle must pick a route without
getting blocked by others, where dedicated parking spaces are included in the layout to
let vehicles make room. Notably, the regular production process runs periodically, so that
the given transport tasks have to be repeatedly executed in ﬁxed intervals.
To make the execution of transport tasks more eﬀective, we propose to perform task
assignment and vehicle routing by means of Answer Set Programming (ASP; Lifschitz
1999), given its declarative and elaboration tolerant approach to combinatorial multiobjective problem solving. This results in a transparent and easily adjustable problem
description along with optimal solutions relative to objective criteria. Regarding performance, it also turns out that our approach is eﬃcient enough to meet the industrial-scale
requirements of production processes at the car factory of Mercedes-Benz Ludwigsfelde
GmbH.
The paper is structured as follows. In the next section, we start by formalizing automated guided vehicle routing for transport tasks evolving in production processes at the
car factory of Mercedes-Benz Ludwigsfelde GmbH. Once this is accomplished, we provide
in Section 3 a corresponding ASP encoding, incorporating the hard and soft constraints
on solutions. In Section 4, we empirically evaluate our approach on use cases designed to
test the practical applicability of control systems for automated guided vehicles. Finally,
we conclude the paper with a brief discussion of related work and the achieved results.

2 Problem Formalization

In what follows, we formally describe the automated guided vehicle routing scenarios
faced in production processes at the car factory of Mercedes-Benz Ludwigsfelde GmbH,
the conditions on their solutions, as well as objective criteria concerning solution quality
devised together with production engineers at Mercedes-Benz Ludwigsfelde GmbH.
An automated guided vehicle routing scenario is speciﬁed in terms of a directed graph
(V, E), where the nodes V stand for locations of interest and the edges E ⊆ V ×V provide
connections between them, along with a set T of transport tasks. Among the nodes in V ,
we distinguish particular halt and park nodes, given by h(V ) ⊆ V and p(V ) ⊆ V such that
h(V ) ∩ p(V ) = ∅ holds. Moreover, each task t ∈ T has an associated non-empty sequence
s1 , . . . , sm  of subtasks, whose elements are halt nodes, i.e., {s1 , . . . , sm } ⊆ h(V ), and
we write s(t) = m and t[i] = si , for 1 ≤ i ≤ m, to refer to the number of or individual
subtasks, respectively, associated with t. In addition, every edge, halt or park node, and
task is characterized by some positive integer, denoted by d(x) for x ∈ E∪h(V )∪p(V )∪T ,
where d(e) expresses the move duration for a connection e ∈ E, d(v) provides the halt
or park duration for a location v ∈ h(V ) ∪ p(V ), and d(t) constitutes the deadline for
completing a task t ∈ T . Finally, a set C of vehicles usable for transport tasks is given
together with, for each c ∈ C, an initial location l(c) ∈ V such that l(c) = l(c ) when
c ∈ C \ {c}. The vehicles can accomplish transport tasks by taking connections and
halting at the locations of respective subtasks, whose halt durations reﬂect time spent to
operate carried materials (e.g., loading or unloading), while further stops are admitted
at park nodes only, included for allowing a vehicle to wait in case its subsequent route is
temporarily blocked by others.
Example 1
The scenario depicted (in black) in Figure 2 consists of the nodes V = {1, . . . , 7}
and edges E = {(1, 2), (1, 7), (2, 3), (3, 4), (4, 5), (4, 7), (5, 6), (6, 1), (7, 1), (7, 4)}. The halt
nodes h(V ) = {2, 4, 5, 6} are indicated by diamonds in Figure 2, and the park node in
p(V ) = {7} is marked by a square. The two tasks in T = {t1 , t2 } include three subtasks
each, where the respective sequences of halt nodes, t1 [1] = 5, t1 [2] = 4, t1 [3] = 2 and
t2 [1] = 6, t2 [2] = 4, t2 [3] = 2, are listed within red or blue labels, respectively, near
the subtasks. Similarly, the initial locations l(c1 ) = 1 and l(c2 ) = 2 of the vehicles in
C = {c1 , c2 } are indicated by corresponding labels. While Figure 2 does not display the
durations, we assume a uniform move duration d(e) = 4 per connection e ∈ E, a halt
duration d(v) = 3 at each of the four halt nodes v ∈ {2, 4, 5, 6}, and a park duration
d(7) = 2 at the park node 7. Moreover, the deadline for both tasks, t1 and t2 , is given

by d(t1 ) = d(t2 ) = 60 and also omitted in Figure 2 for better clarity.
A solution to a guided vehicle routing scenario comprises a task assignment α : T → C
along with a strict partial order ≺ on T such that either t ≺ t or t ≺ t holds when
α(t) = α(t ) for tasks t = t . In addition, for each vehicle c ∈ C, it contains a route
π(c) = r1 , . . . , rk , where {r1 , . . . , rk } ⊆ E ∪ h(V ) ∪ p(V ), subject to the following
conditions:
1. If ri = (v, v  ), for (v, v  ) ∈ E, or ri = v, for v ∈ h(V ) ∪ p(V ), holds for some 1 ≤ i ≤ k,
then we require that ri−1 = (v  , v) or ri−1 = v, where we let r0 = l(c).

Fig. 2. Optimal routes to accomplish two transport tasks with three subtasks each by two
vehicles

2. For any subtask t[j] of a task t ∈ T with α(t) = c, we deﬁne its completion index by
g(t[j]) = min{i | 1 ≤ i ≤ k, ri = t[j], max({g(t[j − 1]) | 1 < j} ∪ {g(t [s(t )]) | t ≺

t}) < i} and require that g(t[j]) ≤ 1≤i≤g(t[j]) d(ri ) ≤ d(t). In words, the subtasks
of all tasks assigned to a vehicle c have to be completed in the order given by ≺ (as
well as sequences of subtasks) within their respective deadlines, where a subtask is
completed once the route π(c) includes a halt at its location such that all preceding
subtasks have been completed before.
3. If ri = v for a halt node v ∈ h(V ) and 1 ≤ i ≤ k, then we require that g(t[j]) = i for
some subtask t[j] of a task t ∈ T with α(t) = c. That is, halts may be included in the
route of a vehicle c exclusively for completing subtasks of tasks assigned to c.
4. With each x ∈ E ∪ V , we associate a set of occupation times deﬁned as u(c, x) =


{d + 1≤j<i d(rj ) | 1 ≤ i ≤ k, ri = x, 1 ≤ d ≤ d(x)} ∪ { 1≤j≤i d(rj ) | 1 ≤ i ≤
k, ri ∈ E, ri = (v, x)}. For any vehicle c ∈ C \ {c}, the following requirements
check that the routes of c and c do not both lead to a joint location at the same
time, and that c and c do not meet in between nodes by taking connections in
opposite directions: for each location v ∈ V , we require that u(c, v) ∩ u(c , v) = ∅,
while u(c, (v, v  )) ∩ u(c , (v  , v)) = ∅ must hold for each bidirectional connection pair
{(v, v  ), (v  , v)} ⊆ E.
Example 2
The red and blue arrows and labels in Figure 2 indicate the routes of a solution to the
automated guided vehicle routing scenario from Example 1, where the task assignment
is given by α(t1 ) = c1 and α(t2 ) = c2 . As no distinct tasks are assigned to the same
vehicle, we have that the order relation ≺ is empty, while c1 and c2 have to complete the
subtasks of t1 or t2 , respectively, in order. The routes taken by c1 and c2 can be traced
by considering the times of completing route elements, included in labels along nodes
and edges, in increasing order, starting from c1 : 0 and c2 : 0 at the initial locations
l(c1 ) = 1 and l(c2 ) = 2. For vehicle c1 , this yields the route π(c1 ) = (1, 7), 7, (7, 4), (4, 5),
5, (5, 6), (6, 1), (1, 2), (2, 3), (3, 4), 4, (4, 7), (7, 1), (1, 2), 2, where the four stops at halt or
park nodes in h(V ) ∪ p(V ) = {2, 4, 5, 6, 7} are of particular interest. In fact, the inclusion
of the park node 7 at the second position of π(c1 ) delays the subsequent move (7, 4) by
the park duration d(7) = 2 in order to avoid arriving at node 4 at the same time as the

other vehicle c2 . Unlike that, the later stops at the halt nodes 5, 4, and 2 are made to
complete the sequence t1 [1] = 5, t1 [2] = 4, t1 [3] = 2 of subtasks. While π(c1 ) makes c1
revisit the halt nodes t1 [2] = 4 and t1 [3] = 2, a halt would not be admitted on the ﬁrst
arrival because the preceding subtask t1 [1] = 5 or t1 [2] = 4, respectively, has not yet been
completed. Hence, the halt at 5 comes before returning to and halting at 4, and then in
turn at 2 at the very end of the route π(c1 ). As one can check, the sum of durations over
connections as well as halt and park nodes in π(c1 ) is 55, which means that the task t1
gets completed within its deadline d(t1 ) = 60. Regarding the other vehicle c2 , tracing the
blue labels in Figure 2 yields the route π(c2 ) = (2, 3), (3, 4), (4, 5), (5, 6), 6, (6, 1), (1, 7),
(7, 4), 4, (4, 7), (7, 1), (1, 2), 2 for completing the sequence t2 [1] = 6, t2 [2] = 4, t2 [3] = 2
of subtasks. Along this route, c2 visits the halt nodes t2 [2] = 4 and t2 [3] = l(c2 ) = 2 twice,
where halts are in both cases made on the second arrival for completing the subtasks
of t2 in order. Moreover, no stop at the park node 7 is needed to avoid meeting the other
vehicle c1 , and the durations of connections and halt nodes in π(c2 ) sum up to 49, so

that also the task t2 gets completed within its deadline d(t2 ) = 60.
While an automated guided vehicle routing scenario may have plenty feasible solutions,
we apply the following objective criteria, below ordered by signiﬁcance, to distinguish
preferred collections of routes π(ci ) = r1i , . . . , rki  for the vehicles ci ∈ C:
1. The makespan ms of a solution is the maximum sum of durations over the route
elements of some vehicle, i.e.,

ms = max{ 1≤ji ≤ki d(rji ) | ci ∈ C}.
2. The route length rl of a solution is the sum of durations over elements in the routes
of all vehicles, i.e.,

rl = ci ∈C,1≤ji ≤ki d(rji ).
3. The crossing number cn is the number of pairs ({ci , ci }, v) such that ci ∈ C, ci ∈
C \ {ci }, and we have that rji = (v  , v) and rji = (v  , v) for some connections
{(v  , v), (v  , v)} ⊆ E with v  = v  , 1 ≤ ji ≤ ki , and 1 ≤ ji ≤ ki , i.e.,
cn = |{({ci , ci }, v) | ci ∈ C, ci ∈ C \ {ci }, 1 ≤ ji ≤ ki , 1 ≤ ji ≤ ki ,
{rji , rji } ⊆ E, rji = (v  , v), rji = (v  , v), v  = v  }|.
In words, the crossing number estimates how often the routes of distinct vehicles
come together at a joint location (at diﬀerent times). Although solutions are such
that vehicles do not share any location at the same time, it is desirable to make
routes as disjoint as possible to keep knock-on eﬀects in case of a disruption short.
4. The overlap number on is the number of triples ({ci , ci }, rji , rji ) such that ci ∈ C,
ci ∈ C \ {ci }, and we have that rji = (v, v  ) and rji = (v, v  ) (or rji = (v  , v)) for
some connection (v, v  ) ∈ E (or connections {(v, v  ), (v  , v)} ⊆ E), 1 ≤ ji ≤ ki , and
1 ≤ ji ≤ ki , i.e.,
on = |{({ci , ci }, rji , rji ) | ci ∈ C, ci ∈ C \ {ci }, 1 ≤ ji ≤ ki , 1 ≤ ji ≤ ki ,
{rji , rji } ⊆ E, rji = (v, v  ), rji ∈ {(v, v  ), (v  , v)}}|.
The motivation for aiming at a small overlap number is the same as with the crossing
number, while connections taken by distinct vehicles are considered instead of crossing
locations. Note that, for a bidirectional connection pair {(v, v  ), (v  , v)} ⊆ E, two

vehicles ci and ci that move between v and v  in either direction can contribute
one, two, or four triples ({ci , ci }, rji , rji ), depending on whether neither, one, or
both include (v, v  ) as well as (v  , v) in their routes. Unlike that, only the triple
/ E.
({ci , ci }, (v, v  ), (v, v  )) is obtained when (v, v  ) ∈ E but (v  , v) ∈
For each of the four objective criteria speciﬁed above, a smaller value is preferred to
greater ones. The order of signiﬁcance is chosen such that the completion of all transport
tasks in as little time as possible has the highest priority, minimizing the overall utilization
of vehicles comes second, keeping routes disjoint is third, and the avoidance of overlapping
connections takes the lowest priority.
Example 3
The solution to the automated guided vehicle routing scenario from Example 1 indicated
in Figure 2 and described further in Example 2 happens to be optimal relative to the
applied objective criteria. Given that the sums 55 and 49 are obtained for the durations
of elements of the route π(c1 ) or π(c2 ), respectively, the makespan ms matches the
maximum 55, and the route length rl amounts to 55 + 49 = 104. Regarding the crossing
number cn, π(c1 ) and π(c2 ) visit three locations, 1, 4, and 7, via diﬀerent connections,
as witnessed by occurrences of (6, 1) and (7, 1), (3, 4) and (7, 4), as well as (1, 7) and
(4, 7) in π(c1 ) or π(c2 ), respectively. Moreover, the overlap number on = 14 is obtained
in view of the inclusion of the connections (1, 2), (2, 3), (3, 4), (4, 5), (5, 6), and (6, 1) in
both π(c1 ) and π(c2 ) along with moves between the nodes 1 and 4 as well as 4 and 7
in either direction, each of the latter contributing four triples counted together by on.
While the gap between the makespan ms = 55 and the deadline d(t1 ) = d(t2 ) = 60 for t1
and t2 may seem small, let us mention that 561 solutions are feasible and that only one
of them is optimal. Such discrepancy clearly indicates that human engineers will hardly
be able to perform an exhaustive optimization of routes in realistic scenarios of greater
size only by hand.

3 Problem Encoding
Following the common modeling methodology of ASP, we represent (optimal) solutions
to automated guided vehicle routing scenarios by facts specifying an instance along
with a uniform problem encoding. To begin with, Table 1 surveys the format of facts
describing the locations, connections, tasks, and vehicles belonging to an automated
guided vehicle routing scenario. The respective fact representation of the scenario from
Example 1 is given in Listing 1, using the shorthands ‘ .. ’ and ‘;’ of clingo (Gebser et al.
2015) to abbreviate facts for a range or collection of arguments, respectively. Note that
facts of the form stay (v ,d(v)) combine halt and park nodes v ∈ h(V ) ∪ p(V ) whose
associated duration d(v) is greater than one, as such non-atomic durations are subject
to dedicated conditions in our problem encoding below. For locations v ∈ V that have
several incoming connections, facts of the form less (v  ,v  ,v ) provide lexicographically
consecutive locations v  < v  such that {(v  , v), (v  , v)} ⊆ E, based on the standard term
order of the ASP-Core-2 language (Calimeri et al. 2012). Our encoding below makes use of
the lexicographical order of predecessor locations for a compact formulation of conditions
to detect crossings. Moreover, facts time (n) give time points of interest, ranging from
zero to the latest deadline of any transport task, and tasks (t ,t ) holds for distinct tasks

Table 1. Fact format for specifying automated guided vehicle routing scenarios in ASP

Fact
node ( v )
halt ( v , d(v) )
park ( v , d(v) )
stay ( v , d(v) )
edge ( v , v  , d(v, v  ) )
less ( v  , v  , v )
time ( n )
task ( t )
task ( t , d(t) )
tasks ( t , t )
subtask ( t ,s( i ))
subtask ( t ,s( i ), v )
vehicle ( c )
vehicle ( c , v )

Meaning
v∈V
v ∈ h(V ) with halt duration d(v)
v ∈ p(V ) with park duration d(v)
v ∈ h(V ) ∪ p(V ) with halt or park duration d(v) > 1
(v, v  ) ∈ E with move duration d((v, v  ))
{(v  , v), (v  , v)} ⊆ E for lexicographically consecutive locations v  < v 
0 ≤ n ≤ max{d(t) | t ∈ T }
t∈T
t ∈ T with deadline d(t)
t ∈ T and t ∈ T \ {t}
t ∈ T with 1 ≤ i ≤ s(t)
t ∈ T , 1 ≤ i ≤ s(t), and t[i] = v with v ∈ h(V )
c∈C
c ∈ C with initial location l(c) = v

Listing 1. Instance of automated guided vehicle routing with two tasks and two vehicles
node(v (1..7)).
halt(v(2),3). stay(v(2),3).
halt(v(4),3). stay(v(4),3).
halt(v(5),3). stay(v(5),3).
halt(v(6),3). stay(v(6),3).
park(v(7),2). stay(v(7),2).
edge(v(6;7),v(1),4). less(v(6),v(7),v(1)).
edge(v(1),v(2),4).
edge(v(2),v(3),4).
edge(v(3;7),v(4),4). less(v(3),v(7),v(4)).
edge(v(4),v(5),4).
edge(v(5),v(6),4).
edge(v(1;4),v(7),4). less(v(1),v(4),v(7)).
time (0..60).
task(t(1)). task(t(1),60). tasks(t(1),t(2)).
task(t(2)). task(t(2),60). tasks(t(2),t(1)).
subtask(t(1),s(1)). subtask (t(1),s(1),v(5)).
subtask(t(1),s(2)). subtask (t(1),s(2),v(4)).
subtask(t(1),s(3)). subtask (t(1),s(3),v(2)).
subtask(t(2),s(1)). subtask (t(2),s(1),v(6)).
subtask(t(2),s(2)). subtask (t(2),s(2),v(4)).
subtask(t(2),s(3)). subtask (t(2),s(3),v(2)).
vehicle(c(1)). vehicle (c(1),v(1)).
vehicle(c(2)). vehicle (c(2),v(2)).

t ∈ T and t ∈ T \ {t}, which have to be ordered by ≺ when α(t) = α(t ). Facts of the
remaining predicates list further properties of automated guided vehicle routing scenarios
in a straightforward way, and their respective meanings are summarized in Table 1.

Table 2. Atoms characterizing solutions to automated guided vehicle routing scenarios
Atom

Meaning

assign ( c , t )
share ( t , t )
order ( t , t )
check ( t )
check ( t , t )

α(t) = c for t ∈ T and c ∈ C
α(t) = α(t ) for {t, t } ⊆ T such that t < t lexicographically
t ≺ t for t ∈ T and t ∈ T \ {t} such that α(t) = α(t )
t ∈ T is topological in (T, {(t , t ) | order ( t , t ) holds})
t ⊀ t or check ( t ) holds for t ∈ T and t ∈ T \ {t}

pass ( c , t ,s( i ), v , d(v) , n )

α(t) = c, 1 ≤ i ≤ s(t), t[i] = v, and n ∈ u(c, v) ∪ {0 | l(c) = v}
for t ∈
T and c ∈ C with π(c) = r1 , . . . , rk  such that
m ≤ 1≤j≤m d(rj ) ≤ n < d(t),
where m = max({g(t[i − 1]) | 1 < i} ∪ {g(t [s(t )]) | t ≺ t})
t ∈ T , 1 ≤ i ≤ s(t), and 1 ≤ n ≤ d(t) such that g(t[i]) ≤

1≤j≤g(t[i]) d(rj ) ≤ n for α(t) = c with π(c) = r1 , . . . , rk 

t ≺ t and 0 ≤ n < min{d(t), d(t
)} for t ∈ T and t ∈ T \ {t}
such that n < max{g(t [s(t )]), 1≤j≤g(t [s(t )]) d(rj )},
where α(t) = α(t ) = c with π(c) = r1 , . . . , rk 

done ( t ,s( i ), n )
wait ( t , n )

at ( c , v , n )
move ( c , v , v  , n )
move ( c , n )
moving ( c , v , v  , n )
moving ( v , v  , n )
free ( c , d(v) , n )

free ( c , n )

used ( c , n )
move ( c , v , v  )
mark ( c , v  , v )
same ( c , c , v , v  )

n ∈ u(c, v) ∪ {0 | l(c) = v} for c ∈ C and v ∈ V with
n ≤ max{d(t) | t ∈ T }

π(c) = r
1 , . . . , ri , . . . , rk  for c ∈ C, ri ∈ E, ri = (v, v ),
and n = 1≤j<i d(rj ) with n + d(ri ) ≤ max{d(t) | t ∈ T }
move ( c , v , v  , n ) holds for some connection (v, v  ) ∈ E
π(c) 
= r1 , . . . , ri , . . . , rk 
for c ∈ C, ri ∈ E, ri = (v, v  ),
and 1≤j<i d(rj ) < n ≤ 1≤j≤i d(rj ) ≤ max{d(t) | t ∈ T }
moving ( c , v , v  , n ) holds for some vehicle c ∈ C
c ∈ C, v ∈ h(V ) ∪ p(V ), and 0 ≤ n ≤ max{d(t) | t ∈ T }
with d(v) > 1 and n ∈ u(c, v) ∪ {0 | l(c) = v} such that
/ u(c, v) ∪ {0 | l(c) = v}, n = n + i ∗ d(v) for some
(n − 1) ∈
i ≥ 0, and {n ∈ u(c, e) | e ∈ E, n < n < n,
n + d(e) − 1 ≤ max{d(t) | t ∈ T }} = ∅
c∈C
 and 0 ≤ n ≤ max{d(t) | t ∈ T } such that
n∈
/ v∈h(V )∪p(V ),d(v)>1 u(c, v) or free ( c , d , n ) holds,
where d = d(v) > 1 for some v ∈ h(V ) ∪ p(V )

c ∈ C and 0 ≤ n ≤ max({0} ∪ v∈V u(c, v)) with
n ≤ max{d(t) | t ∈ T }
move ( c , v , v  , n ) holds for some 0 ≤ n ≤ max{d(t) | t ∈ T }
move ( c , v  , v ) holds for some connections
{(v  , v), (v  , v)} ⊆ E such that v  < v  lexicographically
move ( c , v , v  ) and move ( c , v , v  ) hold for {c, c } ⊆ C
such that c < c lexicographically

Our uniform problem encoding, given in Listings 2 and 3, can be understood as a
merger of four logical parts, addressing task assignment and ordering, task completion,
vehicle routing, as well as objective criteria (the latter part shown separately in Listing 3).
Table 2 follows this logical structure in providing the meanings of predicates deﬁned in
each of the four parts, where a condition written in the right column applies if and only
if a corresponding atom of the form given in the left column belongs to a stable model.
In more detail, the encoding part addressing task assignment and ordering is shown in
Lines 3–12 of Listing 2. The choice rule in Line 3 makes sure that exactly one atom of the

Listing 2. Encoding of task assignment, completion, and automated guided vehicle routing
1

% task assignment

3

{assign(C,T) : vehicle (C)} = 1 :- task(T).

5
6
7

share(T1,T2) :- assign(C,T1), assign(C,T2), T1 < T2.
{order(T1,T2)} :- share(T1,T2).
order(T2,T1) :- share(T1,T2), not order(T1,T2).

9
10
11
12

check(T1,T2)
check(T1,T2)
check(T2)
:- task(T2),

14

% task completion

16
17
18
19

pass(C,T,s(1),V,D,N) :- assign(C,T), at(C,V,N), halt(V,D), task(T,M),
subtask (T,s(1),V), not wait(T,N), N < M.
pass(C,T,s(I),V,D,N) :- assign(C,T), at(C,V,N), halt(V,D), task(T,M),
subtask (T,s(I),V), done(T,s(I -1),N), N < M.

21
22
23

done(T,S,N+D) :- pass(C,T,S,V,D,N), task(T,M), not move(C,N), N+D <= M.
done(T,S,N+1) :- done(T,S,N), not task(T,N).
:- task(T,M), subtask (T,S), not done(T,S,M).

25
26
27

wait(T2,N) :- order(T1,T2), task(T1,M1), task(T2,M2), time(N),
subtask (T1,s(I)), not subtask (T1,s(I+1)),
not done(T1,s(I),N), N < M1, N < M2.

29

% vehicle routing

31
32
33
34
35
36

at(C,V,0)
:- vehicle(C,V), time (0).
at(C,V2,N+D) :- move(C,V1,V2,N), edge(V1,V2,D).
at(C,V,N+1) :- at(C,V,N), park(V,D), time(N+1), not move(C,N).
at(C,V,N+1) :- pass(C,T,S,V,D,N), not done(T,S,N), not move(C,N).
:- node(V), time(N), #count{C : at(C,V,N)} > 1.
:- vehicle(C), time(N), #count{V : at(C,V,N)} > 1.

38
39
40

{move(C,V1,V2,N) : edge(V1,V2,D), time(N+D)} < 2 :- vehicle(C), time(N).
move(C,N) :- move(C,V1,V2,N).
:- move(C,V1,V2,N), not at(C,V1,N).

42
43
44
45

moving(C,V1,V2,N+1)
moving(C,V1,V2,N+1)
moving(V1,V2,N)
:- moving(V1,V2,N),

47
48
49
50
51

free(C,D,N)
free(C,D,N+D)
free(C,N)
free(C,N)
:- move(C,N),

:- tasks(T1,T2), not order(T1,T2).
:- tasks(T1,T2), check(T1).
:- task(T2), check(T1,T2) : tasks(T1,T2).
not check(T2).

:- move(C,V1,V2,N).
:- moving(C,V1,V2,N), time(N+1), not at(C,V2,N).
:- moving(C,V1,V2,N).
moving(V2,V1,N).

:- at(C,V,N), stay(V,D), not at(C,V,N -1).
:- free(C,D,N), time(N+D), not move(C,N).
:- free(C,D,N).
:- vehicle (C), time(N), not at(C,V,N) : stay(V,D).
not free(C,N).

form assign (c ,t ) holds per task t ∈ T , providing the vehicle c ∈ C such that α(t) = c.
Further rules deal with the order ≺ among tasks assigned to a common vehicle, where

is derived by the rule in Line 5 if α(t) = α(t ) for tasks t < t . The choice rule
in Line 6 then either picks order (t ,t ), expressing that t ≺ t , or the rule in Line 7 yields
order ( t , t ) , standing for t ≺ t, otherwise. The remaining rules in Lines 9–12 check that
the order relation ≺ is acyclic, utilizing modeling methods detailed in (Gebser et al.).
The second encoding part in Lines 16–27 deals with the completion of transport tasks.
To this end, the two rules in Lines 16–19 indicate time points n such that the vehicle c ∈ C
with α(t) = c visits the halt node given by a subtask t[i], while any preceding subtasks
according to ≺ as well as the sequence of subtasks of t are already completed at time n.
In case i = 1, the latter condition is in Line 17 checked by the absence of yet incomplete
tasks t such that t ≺ t, and otherwise the completion of t[i − 1] up to time n is required
by preconditions in Line 19. The rule in Line 21 then derives done (t ,s(i),n + d(t[i])) ,
signaling the completion of t[i], provided that n+d(t[i]) does not exceed the deadline d(t)
for t and the vehicle c to which t is assigned halts at time n. Note that the occurrence of
a halt in the route of c is determined by the absence of move (c ,n) from a stable model, an
atom that would otherwise indicate that c takes some connection at time n. The condition
that all transport tasks have to be completed within their respective deadlines is further
taken care of by the rule in Line 22, which successively propagates the completion of a
subtask t[i] on to the deadline d(t), along with requiring the completion of any subtask
at d(t) by means of the integrity constraint in Line 23. Finally, the rule in Lines 25–27
derives wait (t ,n), an atom inspected in Line 17, in view of a yet incomplete subtask
t [s(t )] of some task t such that t ≺ t at time n < d(t), where considering the last
subtask t [s(t )] of t only is suﬃcient because the subtasks of t must be completed in
order.
The actual routes of vehicles are tracked by the third encoding part in Lines 31–51. To
begin with, the rule in Line 31 derives at (c ,l(c),0), expressing that any vehicle c ∈ C
starts from its initial location l(c) at time 0. The new location v  ∈ V resulting from
a move (v, v  ) ∈ E at time n is reﬂected by the rule in Line 32 yielding at (c ,v  ,n +
d((v, v  ))). A stop at some park node v ∈ p(V ) is addressed by the rule in Line 33,
deriving at (c ,v ,n + 1) from at (c ,v ,n) in the absence of any move made by c at time n.
Moreover, the rule in Line 34 takes care of a halt to complete some subtask t[i] by deriving
at ( c , t[i] , n + 1 ) in case no subtask preceding t[i] is yet incomplete and t[i] itself is not
already completed at time n, provided that the vehicle c with α(t) = c does not make any
move at n. As a consequence, atoms of the form at (c ,v ,n) in a stable model match times
n ∈ u(c, v) ∪ {0 | l(c) = v} for vehicles c ∈ C and locations v ∈ V , where halts can only
be made for completing assigned subtasks. Given this, the integrity constraint in Line 35
makes sure that u(c, v) ∩ u(c , v) = ∅ for any c ∈ C \ {c}, and the one in Line 36 asserts
that u(c, v)∩u(c, v  ) = ∅ for any v  ∈ V \{v}, thus constituting a redundant/entailed state
constraint that may still be helpful regarding solving performance (Gebser et al. 2012).
Unlike stops at halt or park nodes, a move (v, v  ) ∈ E by some vehicle c ∈ C at
time n such that n + d((v, v  )) is still of interest is signaled by an atom move (c ,v ,v  ,n),
picked by means of the choice rule in Line 38. The rule in Line 39 then provides the
projection of such an atom to move (c ,n), and the integrity constraint in Line 40 requires
c to be at the location v at time n. Moreover, atoms moving (c ,v ,v  ,n ), derived by the
rules in Lines 42–43 for n < n ≤ n + d((v, v  )), yield time points n ∈ u(c, (v, v  )).
Their projection to moving (v ,v  ,n ) in Line 44 is inspected by the integrity constraint
in Line 45 to, in case (v  , v) ∈ E, make sure that u(c, (v, v  )) ∩ u(c , (v  , v)) = ∅ holds for

Listing 3. Encoding part for makespan, route length, crossing, and overlap minimization
53

% objective criteria

55
56
57
58

used(C,N)
:- at(C,V,N).
used(C,N -1) :- used(C,N), 0 < N.
:˜ used(C,N). [1@4,N]
:˜ used(C,N). [1@3,C,N]

60

move(C,V1,V2) :- move(C,V1,V2,N).

62
63
64
65

mark(C,V1,V) :- less(V1,V2,V), move(C,V2,V).
mark(C,V1,V) :- less(V1,V2,V), mark(C,V2,V).
:˜ mark(C1,V1,V), move(C2,V1,V), C1 < C2. [1@2,C1,C2,V]
:˜ move(C1,V1,V), mark(C2,V1,V), C1 < C2. [1@2,C1,C2,V]

67
68
69
70
71
72
73
74
75
76

same(C1,C2,V1,V2) :- move(C1,V1,V2), move(C2,V1,V2),
:˜ move(C1,V1,V2), move(C2,V2,V1), C1 < C2, V1 < V2.
:˜ move(C1,V2,V1), move(C2,V1,V2), C1 < C2, V1 < V2.
:˜ same(C1,C2,V1,V2), V1 < V2.
:˜ same(C1,C2,V2,V1), V1 < V2.
:˜ same(C1,C2,V1,V2), move(C1,V2,V1), V1 < V2.
:˜ same(C1,C2,V1,V2), move(C2,V2,V1), V1 < V2.
:˜ same(C1,C2,V2,V1), move(C1,V1,V2), V1 < V2.
:˜ same(C1,C2,V2,V1), move(C2,V1,V2), V1 < V2.
:˜ same(C1,C2,V1,V2), same(C1,C2,V2,V1), V1 < V2.

C1 < C2.
[1@1,C1,C2,V1,V2]
[1@1,C1,C2,V1,V2]
[1@1,C1,C2,V1,V2]
[1@1,C1,C2,V1,V2]
[1@1,C1,C2,V2,V1]
[1@1,C1,C2,V2,V1]
[1@1,C1,C2,V2,V1]
[1@1,C1,C2,V2,V1]
[2@1,C1,C2,V1,V2]

any c ∈ C \ {c}. The remaining rules in Lines 47–51 check that the moves of a vehicle c
respect non-atomic durations d(v) > 1 of halt and park nodes v ∈ h(V ) ∪ p(V ). To this
end, an atom free (c ,d(v),n) is derived by the rule in Line 47 when c is not already at v
at time n − 1, and then propagated on in steps of the halt or park duration d(v) by the
rule in Line 48 as long as c does not make any move. The projection to free (c ,n) in
Line 49 along with the rule in Line 50, also deriving free (c ,n) in case c is not at any
halt or park node v ∈ h(V ) ∪ p(V ) with d(v) > 1 at time n, indicate all times n at which
c may make a move, and the integrity constraint in Line 51 restricts moves to such time
points.
While the encoding parts described so far make sure that stable models match solutions
to automated guided vehicle routing scenarios, the fourth part in Listing 3 addresses the
objective criteria to distinguish optimal routes. To begin with, the rules in Lines 55–56
derive atoms used (c ,n) to indicate that the route of a vehicle c ∈ C is not ﬁnished before
time n. The makespan ms is then minimized by means of the weak constraint with the
highest priority 4 in Line 57, associating a cost of 1 with each time n. To then minimize
the route length rl with priority 3, the weak constraint in Line 58 likewise penalizes
used ( c , n ) , where a cost of 1 for each time n is charged per vehicle c. The rule in Line 60
provides the projection of moves to move (c ,v ,v  ) in order formulate conditions regarding
crossings and overlaps of routes without referring to particular times of moves. In fact,
the rules in Lines 62–63 investigate and propagate move (c ,v  ,v ) on to locations v  such
that (v  , v) ∈ E and v  < v  lexicographically, thus signaling the inclusion of some other
predecessor location of v in the route of c in terms of mark (c ,v  ,v ). The weak constraints
in Lines 64–65 make use of this to detect a crossing at v by the move (v  , v) of another

vehicle c ∈ C \ {c}, while the lexicographical order of distinct predecessor locations of v
in the routes of c and c may likewise be switched. In either case, a crossing at v accounts
for a cost of 1 with priority 2, where the set {c, c } in pairs ({c, c }, v) contributing
to cn is reﬂected by requiring that c < c lexicographically. The remaining rules and
weak constraints take care of overlapping connections, where the rule in Line 67 yields
same ( c , c , v , v  ) when distinct vehicles c < c both include (v, v  ) in their routes. This
case as well as taking (v, v  ) and (v  , v) in opposite directions is sanctioned by a cost of 1
with priority 1 in view of the weak constraints in Lines 68–71. If c or c includes both
directions, (v, v  ) and (v  , v), in its route, an additional cost of 1 is charged by some of
the weak constraints in Lines 72–75, by means of using a diﬀerent term order regarding v
and v  than before. Finally, the weak constraint in Line 76 accounts for a cost of 2 if c and
c both take (v, v  ) as well as (v  , v), in which case the resulting sum of four matches the
number of triples ({c, c }, e, e ) such that {e, e } = {(v, v  ), (v  , v)} contributing to on. As
a consequence, we have that the sums of costs with particular priority coincide with the
objectives ms, rl , cn, and on, where priorities follow the criteria’s order of signiﬁcance.
Example 4
The optimal solution to the automated guided vehicle routing scenario from Example 1
indicated in Figure 2 and described further in Example 2 is such that the vehicles c1
and c2 ﬁnish their last halts at t1 [3] = t2 [3] = 2 at the times 55 and 49. Hence, a
stable model representing this solution, obtained for the facts in Listing 1 along with the
encoding in Listings 2 and 3, contains the atoms at (c (1) ,v (2) , 55) and at (c (2) ,v (2) , 49) .
These in turn imply atoms used (c(i),ni ) for i ∈ {1, 2}, 0 ≤ n1 ≤ 55, and 0 ≤ n2 ≤
49. The weak constraints with priorities 4 and 3 thus yield 56 or 106 distinct terms,
respectively, of the form [1 @4 ,ni ] or [1 @3 ,c(i),ni ], where the numbers of terms match
ms + 1 = 55 + 1 = 56 and rl + i = 104 + 2 = 106. Note that the addends 1 or i to
ms and rl are due to atoms of the form used (c(i), 0) , which can be deterministically
derived in view of the initial locations of vehicles and do not aﬀect the preference order
of solutions. Regarding crossings, as detailed in Example 3, the connections taken by
c1 and c2 trigger weak constraints associated with the terms [1 @2 ,c (1) ,c (2) ,v (1)] ,
[1 @2 ,c (1) ,c (2) ,v (4)] , and [1 @2 ,c (1) ,c (2) ,v (7)] , leading to the sum cn = 3 of costs.
Moreover, six unidirectional overlaps of the routes of c1 and c2 are indicated by terms of
the form [1 @1 ,c (1) ,c (2) ,v(v ),v(v  )] for (v, v  ) ∈ {(1, 2), (2 , 3), (3 , 4), (4 , 5), (5 , 6), (1, 6)},
among which the lexicographically ordered pair (1, 6) stands for the connection (6, 1)
taken in the opposite direction. However, such term ordering matters for bidirectional
connections only, where (v, v  ) ∈ {(1, 4), (4 , 7)} yield three distinct terms each: [1 @1 ,
c (1) ,c (2) ,v( v ),v( v  )] , [1 @1 ,c (1) ,c (2) ,v( v  ),v( v )] , and [2 @1 ,c (1) ,c (2) ,v( v ),v( v  )] .
That is, the weak constraints with priority 1 lead to twelve terms in total, ten of them
accounting for a cost of 1 and two having a cost of 2 (written before ‘@’), summing up to
on = 14.


4 Empirical Results
A preliminary ﬁeld study by production engineers at Mercedes-Benz Ludwigsfelde GmbH
investigated viable options for implementing a new control system for automated guided
vehicles used in car assembly, where one goal consists of setting the routes for such

Fig. 3. Use cases solved by means of ASP (in blue) and the default scheduler of openTCS (in
red)

vehicles up automatically rather than by hand. This study raised interest in the opensource control system software openTCS (see http://www.opentcs.org), developed by
Fraunhofer IML. While openTCS oﬀers a generic platform for automated guided vehicle
control, it is not geared to ship a solver for hard combinatorial tasks such as the multiobjective optimization of routes oﬀ the shelf. Instead, the system is extensible and allows
for integrating customized components, which is where our ASP approach to vehicle
routing is envisaged to come into play in the future. For reference, however, we contrast
our results to those obtained with the default scheduler of openTCS, which implements
a simple round-robin procedure to assign tasks to and pick shortest routes for vehicles.
In order to evaluate the practical applicability of control systems, the engineers at
Mercedes-Benz Ludwigsfelde GmbH deﬁned 18 use cases based on a factory layout
with 25 locations and 35 connections, resembling the storage areas, assembly lines,
and transport routes at the physical car factory. Such use cases are run in simulation to test the entire functionality of a control system, where task assignment and
vehicle routing are incorporated at the high level. According to the main test targets,
the use cases (available at http://www.cs.uni-potsdam.de/wv/projects/daimler/
resources-iclp18.tar.xz) are grouped into ﬁve categories: A) communication and
feedback, B) task assignment and execution, C) routing and traﬃc management, D)
special conditions and breakdowns, and E) full production cycle. While test cases in the
ﬁrst four groups focus on particular scenarios of small size, i.e., up to three tasks and
vehicles, the use case in category E emulates a full production cycle of roughly 20 minutes
in real time, involving ten transport tasks with 39 subtasks in total to be accomplished
by four automated guided vehicles.
For computing optimal routes with clingo (version 5.2.2), we represented the scenarios
of the use cases in terms of facts as described in the previous section, thus focusing on
task assignment and vehicle routing rather than system control. As indicated by the
blue bars in Figure 3, we succeeded in obtaining provably optimal routes in all of the
18 use cases, where the runtimes of clingo on a desktop machine equipped with Intel i76700 3.40GHz CPU ranged from split seconds to ten seconds at most for each of the 17
small scenarios in the ﬁrst four groups. While these small instances stem from use cases

primarily created to test the reaction of a control system to some isolated situation,
the scenario in E aims at managing all of the transport tasks recurring within each
full production cycle, a complex operation that has so far been carried out by human
engineers before a production process starts or is resumed with an updated conﬁguration,
respectively. In fact, optimizing the routes for automated guided vehicles over the full
production cycle is computationally challenging, even for a system like clingo that is
geared to hard combinatorial problem solving, and it took about 5 hours to compute
a provably optimal solution. Its makespan of 225 clock cycles amounts to ﬁve times as
many seconds in real time, roughly 20 minutes, the route length 891 corresponds to about
75 minutes of operation, shared between four vehicles whose computed routes involve 39
crossings and 141 overlaps in total. While 5 hours of computation time can certainly not
be aﬀorded by a control system that has to react to incidents in real time, the dimension
of what has to be anticipated in comparable scenarios is nevertheless encouraging, as it
tells us that routes for accomplishing recurrent transport tasks in realistic production
processes can be programmed in an automated fashion, going along with optimality that
cannot be established by human engineers alone.
The red bars shown in Figure 3 exhibit that the default scheduler shipped with
openTCS fails to come up with a feasible solution in 8 of the 18 use cases, which is
due to its greedy approach to assign a task and lock locations along the shortest route of
a vehicle one by one. E.g., such an approach is bound to fail on the example scenario in
Figure 2, where each of the vehicles c1 and c2 must necessarily visit node 4 in order to
complete any subtask, which locks the respective other vehicle out from visiting node 4,
while a single vehicle cannot alone complete both transport tasks, t1 and t2 , within their
deadlines. This observation along with the fact that clingo can timely handle the small
scenarios of use cases in the ﬁrst four groups clearly motivate its integration into openTCS
as a component in charge of task assignment and vehicle routing, which we are working
on.
5 Discussion
The problem of assigning transport tasks to and routing automated guided vehicles
in car assembly reveals parallels to Generalized Target Assignment and Path Finding
(GTAPF; Nguyen et al. 2017) as well as Temporal Planning (cf. Fisher 2008). Similar to
automated guided vehicle routing scenarios, GTAPF allows for more tasks than there are
agents, sequences of subtasks, and deadlines for tasks, while its framework assumes that
tasks are completed according to some total order and it does not feature (non-atomic)
durations. Unlike that, durations play a fundamental role in Temporal Planning, but
standard representations such as the Planning Domain Deﬁnition Language (PDDL; Fox
and Long 2003) lack constructs for conveniently modeling task assignment and ordering.
We thus stick to a native ASP encoding, and a corresponding approach likewise turned
as advantageous for planning by means of tabled logic programming (Zhou et al. 2015).
Early work on using ASP for solving basic multi-agent path ﬁnding problems was done
by Erdem et al. (2013); also, Nguyen et al. (2017) used ASP for solving GTAPF problems.
Action durations and intervals were previously considered in ASP by Son et al. (2004).
Notably, we checked that clingo requires less than 1 GB of RAM to represent durations
and makespan of a full production cycle at the car factory of Mercedes-Benz Ludwigsfelde

GmbH, while extensions by diﬀerence logic (Neubauer et al. 2017) or integer variables
(Banbara et al. 2017) may possibly handle even greater time intervals in a compact way.
The contributions of our work include a transparent and elaboration tolerant formalization of transport tasks evolving in realistic production processes by specifying them
in ASP, thus also making oﬀ-the-shelf solving systems accessible for computing optimal
vehicle routes. As it turns out, our declarative approach allows for handling a scenario
covering the full production cycle at the car factory of Mercedes-Benz Ludwigsfelde
GmbH, so that it can assist human engineers in setting up routes to be periodically
taken by automated guided vehicles. Moreover, scenarios of small size, resembling task
reassignment or vehicle rerouting in case of unforeseen circumstances, can virtually be
handled in real time, which also makes the future integration of clingo as a component of
the control system software openTCS for task assignment and vehicle routing attractive.

